Всем доброго дня!
Уважаемые форумчане, помогите найти ошибку в запросе. Конфигурация самописная, на БСП 3.1.6. Платформа 8.3.22.1923. В конфигурации справочник ФизическиеЛица с табличной частью КонтакнаяИнформация и регистр накопления ДоговорныеОбязательства. К ним обращен запрос:

```1c
    ВЫБРАТЬ РАЗРЕШЕННЫЕ РАЗЛИЧНЫЕ
    ДоговорныеОбязательства.Договор КАК Договор,
    ДоговорныеОбязательства.Абонент КАК Абонент
    ПОМЕСТИТЬ ВТ_Абоненты
    ИЗ
    РегистрНакопления.ДоговорныеОбязательства КАК ДоговорныеОбязательства
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ РАЗРЕШЕННЫЕ РАЗЛИЧНЫЕ
    ФизическиеЛицаКонтактнаяИнформация.Ссылка КАК Абонент,
    ЕСТЬNULL(ФизическиеЛицаКонтактнаяИнформация.НомерТелефона, "нет") КАК НомерТелефона,
    ВТ_Абоненты.Договор КАК Договор
    ИЗ
    ВТ_Абоненты КАК ВТ_Абоненты
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ФизическиеЛица.КонтактнаяИнформация КАК ФизическиеЛицаКонтактнаяИнформация
    ПО (ВТ_Абоненты.Абонент = ФизическиеЛицаКонтактнаяИнформация.Ссылка)
    ГДЕ
    ФизическиеЛицаКонтактнаяИнформация.Тип = &ТипКИ

    УПОРЯДОЧИТЬ ПО
    Абонент
```

Как я полагал, применение ЕСТЬNULL(ФизическиеЛицаКонтактнаяИнформация.НомерТелефона, "нет") и левого соединения при объединении таблиц дадут мне на выходе все записи (различные по Абонент+Договор) из регистра ДоговорныеОбязательства и контактную информация из справочника ФизическиеЛица, отсутствующая конт.информация будет заменена на "нет".
Но так не получается. Если нужная (ФизическиеЛицаКонтактнаяИнформация.Тип = &ТипКИ) отсутствует, строка не выводится совсем.
В чем я неправ, подскажите?
---
Надо таблицу ФизическиеЛицаКонтактнаяИнформация связать левым соединением с ВТ_Абоненты, а не наоборот, как у вас. Поспешил, сначала ерунду написал.

Условие на тип КИ в условиях связи укажите.
---
Надо таблицу ФизическиеЛицаКонтактнаяИнформация связать левым соединением с ВТ_Абоненты, а не наоборот, как у вас. Поспешил, сначала ерунду написал.

Условие на тип КИ в условиях связи укажите.
---
Если так:

```1c
    ВЫБРАТЬ РАЗРЕШЕННЫЕ РАЗЛИЧНЫЕ
    ДоговорныеОбязательства.Договор КАК Договор,
    ДоговорныеОбязательства.Абонент КАК Абонент
    ПОМЕСТИТЬ ВТ_Абоненты
    ИЗ
    РегистрНакопления.ДоговорныеОбязательства КАК ДоговорныеОбязательства
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ РАЗРЕШЕННЫЕ РАЗЛИЧНЫЕ
    ФизическиеЛицаКонтактнаяИнформация.Ссылка КАК Абонент,
    ЕСТЬNULL(ФизическиеЛицаКонтактнаяИнформация.НомерТелефона, "нет") КАК НомерТелефона,
    ВТ_Абоненты.Договор КАК Договор
    ИЗ
    Справочник.ФизическиеЛица.КонтактнаяИнформация КАК ФизическиеЛицаКонтактнаяИнформация
    ЛЕВОЕ СОЕДИНЕНИЕ ВТ_Абоненты КАК ВТ_Абоненты
    ПО (ФизическиеЛицаКонтактнаяИнформация.Ссылка = ВТ_Абоненты.Абонент)
    ГДЕ
    ФизическиеЛицаКонтактнаяИнформация.Тип = &ТипКИ

    УПОРЯДОЧИТЬ ПО
    Абонент
```

Результат тот-же.
---
Красавчик! Тебе сказали добавить еще одну связь, а ты просто поменял местами две таблицы?
Вот как тебе что-то советовать?
---
```1c
    ВЫБРАТЬ РАЗРЕШЕННЫЕ РАЗЛИЧНЫЕ
    ДоговорныеОбязательства.Договор КАК Договор,
    ДоговорныеОбязательства.Абонент КАК Абонент
    ПОМЕСТИТЬ ВТ_Абоненты
    ИЗ
    РегистрНакопления.ДоговорныеОбязательства КАК ДоговорныеОбязательства
    ;

    ////////////////////////////////////////////////////////////­­////////////////////
    ВЫБРАТЬ РАЗРЕШЕННЫЕ РАЗЛИЧНЫЕ
    ВТ_Абоненты.Абонент КАК Абонент,
    ВТ_Абоненты.Договор КАК Договор,
    ЕСТЬNULL(ФизическиеЛицаКонтактнаяИнформация.НомерТелефона, "нет") КАК НомерТелефона
    ИЗ
    ВТ_Абоненты КАК ВТ_Абоненты
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ФизическиеЛица.КонтактнаяИнформация КАК ФизическиеЛицаКонтактнаяИнформация
    ПО (ВТ_Абоненты.Абонент = ФизическиеЛицаКонтактнаяИнформация.Ссылка
    И ФизическиеЛицаКонтактнаяИнформация.Тип = &ТипКИ)

    УПОРЯДОЧИТЬ ПО
    Абонент
```
---
Еще раз спасибо!!!
---
Спасибо, так лучше. Результат на все 100%. Нашел подтверждение в инете. Условие на ТипКИ приводит к выбрасыванию строки ведущей(левой) таблицы из результата. А вот условие на закладке "Связи" дает тот результат который мне и нужен. Спасибо что поняли меня. Правильно заданный вопрос приводит к правильному ответу.
---
Если нужная (ФизическиеЛицаКонтактнаяИнформация.Тип = &ТипКИ) отсутствует, строка не выводится совсем. А должна?
---
Хотелось-бы, мне нужны все договора и физические лица у которых есть телефон, а если нет - должно выводится слово "нет".
---
Хотеть в нашей стране не запрещено. Пока ещё.
Вопрос в другом - ты сам понимаешь, что ты написал?